

power twovariances — Power analysis for a two-sample variances test

Description	Quick start	Menu	Syntax
Options	Remarks and examples	Stored results	Methods and formulas
References	Also see		

Description

`power twovariances` computes sample size, power, or the experimental-group variance (or standard deviation) for a two-sample variances test. By default, it computes sample size for given power and the values of the control-group and experimental-group variances. Alternatively, it can compute power for given sample size and values of the control-group and experimental-group variances or the experimental-group variance for given sample size, power, and the control-group variance. Also see [PSS-2] [power](#) for a general introduction to the `power` command using hypothesis tests.

Quick start

Sample size for a test of $H_0: \sigma_1^2 = \sigma_2^2$ versus $H_a: \sigma_1^2 \neq \sigma_2^2$ with control-group variance $v_1 = 25$, experimental-group variance $v_2 = 36$, default power of 0.8, and significance level $\alpha = 0.05$

```
power twovariances 25 36
```

Same as above, but specified as standard deviations $s_1 = 5$ and $s_2 = 6$

```
power twovariances 5 6, sd
```

Sample size for $v_1 = 25$ and v_2 equals to 36, 38, 40, and 42

```
power twovariances 25 (36(2)42)
```

Same as above, but display results in a graph of sample size versus v_2

```
power twovariances 25 (36(2)42), graph
```

Save results to the dataset `mydata.dta`

```
power twovariances 25 (36(2)42), saving(mydata)
```

Power for a total sample size of 300

```
power twovariances 25 36, n(300)
```

Same as above, but specify sample sizes of 200 and 100 for groups 1 and 2, respectively

```
power twovariances 25 36, n1(200) n2(100)
```

Effect size and experimental-group standard deviation given control-group standard deviation of 5, sample size of 200, and power of 0.8

```
power twovariances 5, sd n(200) power(0.8)
```

Same as above, but calculate experimental-group variance given control-group variance of 25

```
power twovariances 25, n(200) power(0.8)
```

Menu

Statistics > Power, precision, and sample size

Syntax

Compute sample size

Variance scale

```
power twovariances  $v_1$   $v_2$  [ , power(numlist) options ]
```

Standard deviation scale

```
power twovariances  $s_1$   $s_2$  , sd [ power(numlist) options ]
```

Compute power

Variance scale

```
power twovariances  $v_1$   $v_2$  , n(numlist) [ options ]
```

Standard deviation scale

```
power twovariances  $s_1$   $s_2$  , sd n(numlist) [ options ]
```

Compute effect size and target parameter

Experimental-group variance

```
power twovariances  $v_1$  , n(numlist) power(numlist) [ options ]
```

Experimental-group standard deviation

```
power twovariances  $s_1$  , sd n(numlist) power(numlist) [ options ]
```

where v_1 and s_1 are the variance and standard deviation, respectively, of the control (reference) group and v_2 and s_2 are the variance and standard deviation of the experimental (comparison) group. Each argument may be specified either as one number or as a list of values in parentheses (see [U] [11.1.8 numlist](#)).

<i>options</i>	Description
<code>sd</code>	request computation using the standard deviation scale; default is the variance scale
Main	
* <code>alpha(numlist)</code>	significance level; default is <code>alpha(0.05)</code>
* <code>power(numlist)</code>	power; default is <code>power(0.8)</code>
* <code>beta(numlist)</code>	probability of type II error; default is <code>beta(0.2)</code>
* <code>n(numlist)</code>	total sample size; required to compute power or effect size
* <code>n1(numlist)</code>	sample size of the control group
* <code>n2(numlist)</code>	sample size of the experimental group
* <code>nratio(numlist)</code>	ratio of sample sizes, $N2/N1$; default is <code>nratio(1)</code> , meaning equal group sizes
<code>compute(N1 N2)</code>	solve for $N1$ given $N2$ or for $N2$ given $N1$
<code>nfractional</code>	allow fractional sample sizes
* <code>ratio(numlist)</code>	ratio of variances, v_2/v_1 (or ratio of standard deviations, s_2/s_1 , if option <code>sd</code> is specified); specify instead of the experimental-group variance v_2 (or standard deviation s_2)
<code>direction(upper lower)</code>	direction of the effect for effect-size determination; default is <code>direction(upper)</code> , which means that the postulated value of the parameter is larger than the hypothesized value
<code>onesided</code>	one-sided test; default is two sided
<code>parallel</code>	treat number lists in starred options or in command arguments as parallel when multiple values per option or argument are specified (do not enumerate all possible combinations of values)
Table	
<code>[no]table [(tablespec)]</code>	suppress table or display results as a table; see [PSS-2] power, table
<code>saving(filename [, replace])</code>	save the table data to <i>filename</i> ; use <code>replace</code> to overwrite existing <i>filename</i>
Graph	
<code>graph [(graphopts)]</code>	graph results; see [PSS-2] power, graph
Iteration	
<code>init(#)</code>	initial value for sample sizes or experimental-group variance
<code>iterate(#)</code>	maximum number of iterations; default is <code>iterate(500)</code>
<code>tolerance(#)</code>	parameter tolerance; default is <code>tolerance(1e-12)</code>
<code>ftolerance(#)</code>	function tolerance; default is <code>ftolerance(1e-12)</code>
<code>[no]log</code>	suppress or display iteration log
<code>[no]dots</code>	suppress or display iterations as dots
<code>notitle</code>	suppress the title

4 power twovariances — Power analysis for a two-sample variances test

*Specifying a list of values in at least two starred options, or at least two command arguments, or at least one starred option and one argument results in computations for all possible combinations of the values; see [U] 11.1.8 **numlist**. Also see the **parallel** option.

collect is allowed; see [U] 11.1.10 **Prefix commands**.

sd does not appear in the dialog box; specification of **sd** is done automatically by the dialog box selected.

notitle does not appear in the dialog box.

where *tablespec* is

```
column[:label] [column[:label] [...]] [, tableopts]
```

column is one of the columns defined below, and *label* is a column label (may contain quotes and compound quotes).

<i>column</i>	Description	Symbol
alpha	significance level	α
power	power	$1 - \beta$
beta	type II error probability	β
N	total number of subjects	N
N1	number of subjects in the control group	N_1
N2	number of subjects in the experimental group	N_2
nratio	ratio of sample sizes, experimental to control	N_2/N_1
delta	effect size	δ
v1	control-group variance	σ_1^2
v2	experimental-group variance	σ_2^2
s1	control-group standard deviation	σ_1
s2	experimental-group standard deviation	σ_2
ratio	ratio of the experimental-group variance to the control-group variance or ratio of the experimental-group standard deviation to the control-group standard deviation (if sd is specified)	σ_2^2/σ_1^2 σ_2/σ_1
target	target parameter; synonym for v2	
_all	display all supported columns	

Column **beta** is shown in the default table in place of column **power** if specified.

Columns **s1** and **s2** are displayed in the default table in place of the **v1** and **v2** columns when the **sd** option is specified.

Column **ratio** is shown in the default table if specified. If the **sd** option is specified, this column contains the ratio of standard deviations. Otherwise, this column contains the ratio of variances.

Options

sd specifies that the computation be performed using the standard deviation scale. The default is to use the variance scale.

Main

`alpha()`, `power()`, `beta()`, `n()`, `n1()`, `n2()`, `nratio()`, `compute()`, `nfractional`; see [PSS-2] [power](#).

`ratio(numlist)` specifies the ratio of the experimental-group variance to the control-group variance, v_2/v_1 , or the ratio of the standard deviations, s_2/s_1 , if the `sd` option is specified. You can specify either the experimental-group variance v_2 as a command argument or the ratio of the variances in `ratio()`. If you specify `ratio(#)`, the experimental-group variance is computed as $v_2 = v_1 \times \#$. This option is not allowed with the effect-size determination.

`direction()`, `onesided`, `parallel`; see [PSS-2] [power](#).

Table

`table`, `table()`, `notable`; see [PSS-2] [power](#), [table](#).

`saving()`; see [PSS-2] [power](#).

Graph

`graph`, `graph()`; see [PSS-2] [power](#), [graph](#). Also see the *column* table for a list of symbols used by the graphs.

Iteration

`init(#)` specifies the initial value for the estimated parameter. For sample-size determination, the estimated parameter is either the control-group size n_1 or, if `compute(N2)` is specified, the experimental-group size n_2 . For the effect-size determination, the estimated parameter is the experimental-group variance v_2 or, if the `sd` option is specified, the experimental-group standard deviation s_2 . The default initial values for the variance and standard deviation for a two-sided test are obtained as a closed-form solution for the corresponding one-sided test with the significance level $\alpha/2$. The default initial values for sample sizes for a χ^2 test are obtained from the corresponding closed-form normal approximation.

`iterate()`, `tolerance()`, `ftolerance()`, `log`, `nolog`, `dots`, `nodots`; see [PSS-2] [power](#).

The following option is available with `power twovariances` but is not shown in the dialog box:

`notitle`; see [PSS-2] [power](#).

Remarks and examples

[stata.com](http://www.stata.com)

Remarks are presented under the following headings:

Introduction

Using power twovariances

Computing sample size

Computing power

Computing effect size and experimental-group variance

Testing a hypothesis about two independent variances

This entry describes the `power twovariances` command and the methodology for power and sample-size analysis for a two-sample variances test. See [PSS-2] [Intro \(power\)](#) for a general introduction to power and sample-size analysis and [PSS-2] [power](#) for a general introduction to the `power` command using hypothesis tests.

Introduction

Investigators are often interested in comparing the variances of two populations, such as comparing variances in yields of corn from two plots, comparing variances of stock returns from two companies, comparing variances of the alcohol concentrations from two different yeast strains, and so on. Before conducting the actual study, the investigators need to find the optimal sample size to detect variations that are beyond tolerable limits or industry-wide standards.

This entry describes power and sample-size analysis for the inference about two population variances performed using hypothesis testing. Specifically, we consider the null hypothesis $H_0: \sigma_2^2 = \sigma_1^2$ versus the two-sided alternative hypothesis $H_a: \sigma_2^2 \neq \sigma_1^2$, the upper one-sided alternative $H_a: \sigma_2^2 > \sigma_1^2$, or the lower one-sided alternative $H_a: \sigma_2^2 < \sigma_1^2$.

Hypothesis testing of variances relies on the assumption of normality. If two independent processes are assumed to follow a normal distribution, then the ratio of their sample variances follows an F distribution, and the corresponding test is known as an F test.

The test of variances is equivalent to the test of standard deviations with the null hypothesis $H_0: \sigma_1 = \sigma_2$. The standard deviation test uses the same F test statistic. The only difference between the two tests is the scale or metric of the variability parameters: variances for the variance test and standard deviations for the standard deviation test. In some cases, standard deviations may provide a more meaningful interpretation than variances. For example, standard deviations of test scores or IQ have the same scale as the mean and provide information about the spread of the observations around the mean.

The `power twovariances` command provides power and sample-size analysis for the F test of two-sample variances or standard deviations.

Using power twovariances

`power twovariances` computes sample size, power, or experimental-group variance for a two-sample variances test. All computations are performed for a two-sided hypothesis test where, by default, the significance level is set to 0.05. You may change the significance level by specifying the `alpha()` option. You can specify the `onesided` option to request a one-sided test. By default, all computations assume a balanced- or equal-allocation design; see [PSS-4] **Unbalanced designs** for a description of how to specify an unbalanced design.

In what follows, we describe the use of `power twovariances` in a variance metric. The corresponding use in a standard deviation metric, when the `sd` option is specified, is the same except variances v_1 and v_2 should be replaced with the respective standard deviations s_1 and s_2 . Note that computations using the variance and standard deviation scales yield the same results; the difference is only in the specification of the parameters.

To compute the total sample size, you must specify the control- and experimental-group variances, v_1 and v_2 , respectively, and, optionally, the power of the test in the `power()` option. The default power is set to 0.8.

Instead of the total sample size, you can compute one of the group sizes given the other one. To compute the control-group sample size, you must specify the `compute(N1)` option and the sample size of the experimental group in the `n2()` option. Likewise, to compute the experimental-group sample size, you must specify the `compute(N2)` option and the sample size of the control group in the `n1()` option.

To compute power, you must specify the total sample size in the `n()` option and the control and the experimental-group variances, v_1 and v_2 , respectively.

Instead of the experimental-group variance v_2 , you may specify the ratio v_2/v_1 of the experimental-group variance to the control-group variance in the `ratio()` option when computing sample size or power.

To compute effect size, the ratio of the experimental-group variance to the control-group variance, and the experimental-group variance, you must specify the total sample size in the `n()` option, the power in the `power()` option, the control-group variance v_1 , and, optionally, the direction of the effect. The direction is upper by default, `direction(upper)`, which means that the experimental-group variance is assumed to be larger than the specified control-group value. You can change the direction to be lower, which means that the experimental-group variance is assumed to be smaller than the specified control-group value, by specifying the `direction(lower)` option.

Instead of the total sample size `n()`, you can specify individual group sizes in `n1()` and `n2()`, or specify one of the group sizes and `nratio()` when computing power or effect size. Also see [Two samples](#) in [PSS-4] **Unbalanced designs** for more details.

In the following sections, we describe the use of `power twovariances` accompanied by examples for computing sample size, power, and experimental-group variance.

Computing sample size

To compute sample size, you must specify the control- and the experimental-group variances, v_1 and v_2 , respectively, and, optionally, the power of the test in the `power()` option. A default power of 0.8 is assumed if `power()` is not specified.

► Example 1: Sample size for a two-sample variances test

Consider a study whose goal is to investigate whether the variability in weights (measured in ounces) of bags of potato chips produced by a machine at a plant A, the control group, differs from that produced by a similar machine at a new plant B, the experimental group. The considered null hypothesis is $H_0: \sigma_A = \sigma_B$ versus a two-sided alternative hypothesis $H_a: \sigma_A \neq \sigma_B$ or, equivalently, $H_0: \sigma_A^2 = \sigma_B^2$ versus $H_a: \sigma_A^2 \neq \sigma_B^2$. The standard deviation of weights from plant A is 2 ounces. The standard deviation of weights from the new plant B is expected to be lower, 1.5 ounces. The respective variances of weights from plants A and B are 4 and 2.25. Investigators wish to obtain the minimum sample size required to detect the specified change in variability with 80% power using a 5%-level two-sided test assuming equal-group allocation. To compute sample size for this study, we specify the control- and experimental-group variances after the command name:

```
. power twovariances 4 2.25
Performing iteration ...
Estimated sample sizes for a two-sample variances test
F test
H0: v2 = v1 versus Ha: v2 != v1
Study parameters:
      alpha = 0.0500
      power = 0.8000
      delta = 0.5625
      v1 = 4.0000
      v2 = 2.2500
Estimated sample sizes:
      N = 194
      N per group = 97
```

A total sample of 194 bags, 97 in each plant, must be obtained to detect the specified ratio of variances in the two plants with 80% power using a two-sided 5%-level test.



► Example 2: Standard deviation scale

We can also specify standard deviations instead of variances, in which case we must also specify the `sd` option:

```
. power twovariances 2 1.5, sd
Performing iteration ...
Estimated sample sizes for a two-sample standard-deviations test
F test
H0: s2 = s1 versus Ha: s2 != s1
Study parameters:
      alpha =    0.0500
      power =    0.8000
      delta =    0.7500
      s1 =     2.0000
      s2 =     1.5000
Estimated sample sizes:
      N =         194
N per group =         97
```

We obtain the same sample sizes as in [example 1](#).



► Example 3: Specifying ratio of variances or standard deviations

Instead of the experimental-group variance of 2.25 as in [example 1](#), we can specify the ratio of variances $2.25/4 = 0.5625$ in the `ratio()` option.

```
. power twovariances 4, ratio(0.5625)
Performing iteration ...
Estimated sample sizes for a two-sample variances test
F test
H0: v2 = v1 versus Ha: v2 != v1
Study parameters:
      alpha =    0.0500
      power =    0.8000
      delta =    0.5625
      v1 =     4.0000
      v2 =     2.2500
      ratio =    0.5625
Estimated sample sizes:
      N =         194
N per group =         97
```

The results are identical to those from [example 1](#).

Similarly, instead of the experimental-group standard deviation of 1.5 as in [example 2](#), we can specify the ratio of standard deviations $1.5/2 = 0.75$ in the `ratio()` option and obtain the same results:

```
. power twovariances 2, sd ratio(0.75)
Performing iteration ...
Estimated sample sizes for a two-sample standard-deviations test
F test
H0: s2 = s1 versus Ha: s2 != s1
Study parameters:
    alpha = 0.0500
    power = 0.8000
    delta = 0.7500
    s1 = 2.0000
    s2 = 1.5000
    ratio = 0.7500
Estimated sample sizes:
    N = 194
    N per group = 97
```

◀

▶ Example 4: Computing one of the group sizes

Continuing with [example 1](#), we will suppose that investigators anticipate a sample of 100 bags from plant A and wish to compute the required number of bags from plant B. To compute sample size for plant B using the study parameters of example 1, we use a combination of the `n1()` and `compute(N2)` options.

```
. power twovariances 4 2.25, n1(100) compute(N2)
Performing iteration ...
Estimated sample sizes for a two-sample variances test
F test
H0: v2 = v1 versus Ha: v2 != v1
Study parameters:
    alpha = 0.0500
    power = 0.8000
    delta = 0.5625
    v1 = 4.0000
    v2 = 2.2500
    N1 = 100
Estimated sample sizes:
    N = 194
    N2 = 94
```

A slightly smaller sample of 94 bags is needed from plant B given a slightly larger sample of bags from plant A to achieve the same 80% power as in [example 1](#).

If the sample size for plant B is known a priori, you can compute the sample size for plant A by specifying the `n2()` and `compute(N1)` options.

◀

▷ Example 5: Unbalanced design

By default, `power twovariances` computes sample size for a balanced- or equal-allocation design. If we know the allocation ratio of subjects between the groups, we can compute the required sample size for an unbalanced design by specifying the `nratio()` option.

Continuing with [example 1](#), we will suppose that the new plant B is more efficient and more cost effective in producing bags of chips than plant A. Investigators anticipate twice as many bags from plant B than from plant A; that is, $n_2/n_1 = 2$. We compute the required sample size for this unbalanced design by specifying the `nratio()` option:

```
. power twovariances 4 2.25, nratio(2)
Performing iteration ...
Estimated sample sizes for a two-sample variances test
F test
H0: v2 = v1 versus Ha: v2 != v1
Study parameters:
    alpha = 0.0500
    power = 0.8000
    delta = 0.5625
    v1 = 4.0000
    v2 = 2.2500
    N2/N1 = 2.0000
Estimated sample sizes:
    N = 225
    N1 = 75
    N2 = 150
```

The requirement on the total sample size increases for an unbalanced design compared with the balanced design from [example 1](#). Investigators must decide whether the decrease of 22 from 97 to 75 in the number of bags from plant A covers the cost of the additional 53 ($150 - 97 = 53$) bags from plant B.

Also see [Two samples](#) in [PSS-4] [Unbalanced designs](#) for more examples of unbalanced designs for two-sample tests.

◀

Computing power

To compute power, you must specify the total sample size in the `n()` option and the control- and experimental-group variances, v_1 and v_2 , respectively.

▷ Example 6: Power of a two-sample variances test

Continuing with [example 1](#), we will suppose that the investigators can afford a total sample of 250 bags, 125 from each plant, and want to find the power corresponding to this larger sample size.

To compute the power corresponding to this sample size, we specify the total sample size in the `n()` option:

```
. power twovariances 4 2.25, n(250)
Estimated power for a two-sample variances test
F test
H0: v2 = v1 versus Ha: v2 != v1
Study parameters:
    alpha =    0.0500
      N =     250
N per group =    125
    delta =    0.5625
      v1 =    4.0000
      v2 =    2.2500
Estimated power:
    power =    0.8908
```

With a total sample of 250 bags, 125 per plant, we obtain a power of roughly 89%. ◀

▶ Example 7: Multiple values of study parameters

In this example, we assess the effect of varying the variances of weights obtained from plant B on the power of our study. Continuing with [example 6](#), we vary the experimental-group variance from 1.5 to 3 in 0.25 increments. We specify the corresponding *numlist* in parentheses:

```
. power twovariances 4 (1.5(0.25)3), n(250)
Estimated power for a two-sample variances test
F test
H0: v2 = v1 versus Ha: v2 != v1
```

alpha	power	N	N1	N2	delta	v1	v2
.05	.9997	250	125	125	.375	4	1.5
.05	.9956	250	125	125	.4375	4	1.75
.05	.9701	250	125	125	.5	4	2
.05	.8908	250	125	125	.5625	4	2.25
.05	.741	250	125	125	.625	4	2.5
.05	.5466	250	125	125	.6875	4	2.75
.05	.3572	250	125	125	.75	4	3

The power decreases from 99.97% to 35.72% as the experimental-group variance gets closer to the control-group variance of 4.

For multiple values of parameters, the results are automatically displayed in a table, as we see above. For more examples of tables, see [\[PSS-2\] power, table](#). If you wish to produce a power plot, see [\[PSS-2\] power, graph](#). ◀

Computing effect size and experimental-group variance

Effect size δ for a two-sample variances test is defined as the ratio of the experimental-group variance to the control-group variance, $\delta = v_2/v_1$. If the `sd` option is specified, effect size δ is the ratio of the experimental-group standard deviation to the control-group standard deviation, $\delta = s_2/s_1$.

Sometimes, we may be interested in determining the smallest effect and the corresponding experimental-group variance that yield a statistically significant result for prespecified sample size and power. In this case, power, sample size, and control-group variance must be specified. In addition, you must also decide on the direction of the effect: upper, meaning $v_2 > v_1$, or lower, meaning $v_2 < v_1$. The direction may be specified in the `direction()` option; `direction(upper)` is the default. If the `sd` option is specified, the estimated parameter is the experimental-group standard deviation instead of the variance.

▷ Example 8: Minimum detectable value of the experimental-group variance

Continuing with [example 6](#), we will compute the minimum plant B variance that can be detected given a total sample of 250 bags and 80% power. To find the variance, after the command name, we specify the plant A variance of 4, total sample size `n(250)`, and power `power(0.8)`:

```
. power twovariances 4, n(250) power(0.8)
Performing iteration ...
Estimated experimental-group variance for a two-sample variances test
F test
H0: v2 = v1 versus Ha: v2 != v1; v2 > v1
Study parameters:
      alpha =    0.0500
      power =    0.8000
         N =      250
N per group =    125
       v1 =    4.0000

Estimated effect size and experimental-group variance:
      delta =    1.6573
       v2 =    6.6291
```

We find that the minimum value of the experimental-group variance that would yield a statistically significant result in this study is 6.63, and the corresponding effect size is 1.66.

In this example, we computed the variance assuming an upper direction, or a ratio greater than 1, $\delta > 1$. To request a lower direction, or a ratio less than 1, we can specify the `direction(lower)` option.

◀

Testing a hypothesis about two independent variances

In this section, we demonstrate the use of the `sdtest` command for testing a hypothesis about two independent variances; see [\[R\] sdtest](#) for details.

▷ Example 9: Comparing two variances

Consider the `fuel` dataset analyzed in [\[R\] sdtest](#). Suppose we want to investigate the effectiveness of a new fuel additive on the mileage of cars. We have a sample of 12 cars, where each car was run without the additive and later with the additive. The results of each run are stored in variables `mpg1` and `mpg2`.

```
. use https://www.stata-press.com/data/r18/fuel
```

```
. sdtest mpg1==mpg2
```

```
Variance ratio test
```

Variable	Obs	Mean	Std. err.	Std. dev.	[95% conf. interval]	
mpg1	12	21	.7881701	2.730301	19.26525	22.73475
mpg2	12	22.75	.9384465	3.250874	20.68449	24.81551
Combined	24	21.875	.6264476	3.068954	20.57909	23.17091

```
ratio = sd(mpg1) / sd(mpg2)                                f = 0.7054
H0: ratio = 1                                             Degrees of freedom = 11, 11
Ha: ratio < 1                Ha: ratio != 1                Ha: ratio > 1
Pr(F < f) = 0.2862          2*Pr(F < f) = 0.5725          Pr(F > f) = 0.7138
```

`sdtest` uses the ratio of the control-group standard deviation to the experimental-group standard deviation as its test statistic. We do not have sufficient evidence to reject the null hypothesis of $H_0: \sigma_1 = \sigma_2$ versus the two-sided alternative $H_a: \sigma_1 \neq \sigma_2$; the p -value > 0.5725 .

We use the estimates of this study to perform a sample-size analysis we would have conducted before the study. We assume an equal-group design and power of 80%.

```
. power twovariances 2.73 3.25, sd power(0.8)
```

```
Performing iteration ...
```

```
Estimated sample sizes for a two-sample standard-deviations test
```

```
F test
```

```
H0: s2 = s1 versus Ha: s2 != s1
```

```
Study parameters:
```

```
alpha = 0.0500
power = 0.8000
delta = 1.1905
s1 = 2.7300
s2 = 3.2500
```

```
Estimated sample sizes:
```

```
N = 522
N per group = 261
```

The total sample size required by the test to detect the difference between the two standard deviations of 2.73 in the control group and of 3.25 in the experimental group is 522, 261 for each group, which is significantly larger than the sample of 12 cars in our `fuel` dataset.

◀

Stored results

`power twovariances` stores the following in `r()`:

Scalars

```
r(alpha)      significance level
r(power)      power
r(beta)       probability of a type II error
r(delta)      effect size
r(N)          total sample size
r(N_a)        actual sample size
r(N1)         sample size of the control group
r(N2)         sample size of the experimental group
```

<code>r(nratio)</code>	ratio of sample sizes, N_2/N_1
<code>r(nratio_a)</code>	actual ratio of sample sizes
<code>r(nfractional)</code>	1 if <code>nfractional</code> is specified, 0 otherwise
<code>r(onesided)</code>	1 for a one-sided test, 0 otherwise
<code>r(v1)</code>	control-group variance
<code>r(v2)</code>	experimental-group variance
<code>r(ratio)</code>	ratio of the experimental- to the control-group variances (or standard deviations if <code>sd</code> is specified)
<code>r(separator)</code>	number of lines between separator lines in the table
<code>r(divider)</code>	1 if <code>divider</code> is requested in the table, 0 otherwise
<code>r(init)</code>	initial value for sample sizes, experimental-group variance, or standard deviation
<code>r(maxiter)</code>	maximum number of iterations
<code>r(iter)</code>	number of iterations performed
<code>r(tolerance)</code>	requested parameter tolerance
<code>r(deltax)</code>	final parameter tolerance achieved
<code>r(ftolerance)</code>	requested distance of the objective function from zero
<code>r(function)</code>	final distance of the objective function from zero
<code>r(converged)</code>	1 if iteration algorithm converged, 0 otherwise

Macros

<code>r(type)</code>	<code>test</code>
<code>r(method)</code>	<code>twovariances</code>
<code>r(scale)</code>	<code>variance</code> or <code>standard deviation</code>
<code>r(direction)</code>	<code>upper</code> or <code>lower</code>
<code>r(columns)</code>	displayed table columns
<code>r(labels)</code>	table column labels
<code>r(widths)</code>	table column widths
<code>r(formats)</code>	table column formats

Matrices

<code>r(pss_table)</code>	table of results
---------------------------	------------------

Methods and formulas

Consider two independent samples from a normal population with means μ_1 and μ_2 and variances σ_1^2 and σ_2^2 . The ratio $(s_1/\sigma_1)^2/(s_2/\sigma_2)^2$ has an F distribution with $n_1 - 1$ numerator and $n_2 - 1$ denominator degrees of freedom. s_1^2 and s_2^2 are the sample variances, and n_1 and n_2 are the sample sizes.

Let σ_1^2 and σ_2^2 be the control-group and experimental-group variances, respectively.

A two-sample variances test involves testing the null hypothesis $H_0: \sigma_2^2 = \sigma_1^2$ versus the two-sided alternative hypothesis $H_a: \sigma_2^2 \neq \sigma_1^2$, the upper one-sided alternative $H_a: \sigma_2^2 > \sigma_1^2$, or the lower one-sided alternative $H_a: \sigma_2^2 < \sigma_1^2$.

Equivalently, the hypotheses can be expressed in terms of the ratio of the two variances: $H_0: \sigma_2^2/\sigma_1^2 = 1$ versus the two-sided alternative $H_a: \sigma_2^2/\sigma_1^2 \neq 1$, the upper one-sided alternative $H_a: \sigma_2^2/\sigma_1^2 > 1$, or the lower one-sided alternative $H_a: \sigma_2^2/\sigma_1^2 < 1$.

The following formulas are based on [Dixon and Massey \(1983, 116–119\)](#).

Let α be the significance level, β be the probability of a type II error, and $F_\alpha = F_{n_1-1, n_2-1, \alpha}$ and $F_{n_1-1, n_2-1, \beta}$ be the α th and the β th quantiles of an F distribution with $n_1 - 1$ numerator and $n_2 - 1$ denominator degrees of freedom.

The power $\pi = 1 - \beta$ is computed using

$$\pi = \begin{cases} 1 - F_{n_1-1, n_2-1} \left(\frac{\sigma_1^2}{\sigma_2^2} F_{1-\alpha} \right) & \text{for an upper one-sided test} \\ F_{n_1-1, n_2-1} \left(\frac{\sigma_1^2}{\sigma_2^2} F_{\alpha} \right) & \text{for a lower one-sided test} \\ 1 - F_{n_1-1, n_2-1} \left(\frac{\sigma_1^2}{\sigma_2^2} F_{1-\alpha/2} \right) + F_{n_1-1, n_2-1} \left(\frac{\sigma_1^2}{\sigma_2^2} F_{\alpha/2} \right) & \text{for a two-sided test} \end{cases} \quad (1)$$

where $F_{n_1-1, n_2-1}(\cdot)$ is the cdf of an F distribution with $n_1 - 1$ numerator and $n_2 - 1$ denominator degrees of freedom.

Let $R = n_2/n_1$ denote the allocation ratio. Then $n_2 = R \times n_1$ and power can be viewed as a function of n_1 . Therefore, for sample-size determination, the control-group sample size n_1 is computed first. The experimental-group size n_2 is then computed as $R \times n_1$, and the total sample size is computed as $n = n_1 + n_2$. By default, sample sizes are rounded to integer values; see [Fractional sample sizes](#) in [PSS-4] **Unbalanced designs** for details.

If either n_1 or n_2 is known, the other sample size is computed iteratively from the corresponding power equation in (1).

The initial values for the sample sizes are obtained from closed-form large-sample normal approximations; see, for example, [Mathews \(2010, 68\)](#).

For a one-sided test, the minimum detectable value of the experimental-group variance is computed as follows:

$$\sigma_2^2 = \begin{cases} \sigma_1^2 \frac{F_{n_1-1, n_2-1, 1-\alpha}}{F_{n_1-1, n_2-1, \beta}} & \text{for an upper one-sided test} \\ \sigma_1^2 \frac{F_{n_1-1, n_2-1, \alpha}}{F_{n_1-1, n_2-1, 1-\beta}} & \text{for a lower one-sided test} \end{cases} \quad (2)$$

For a two-sided test, the minimum detectable value of the experimental-group variance is computed iteratively using the two-sided power equation from (1). The default initial value is obtained from (2) with α replaced by $\alpha/2$.

References

- Dixon, W. J., and F. J. Massey, Jr. 1983. *Introduction to Statistical Analysis*. 4th ed. New York: McGraw-Hill.
- Mathews, P. 2010. *Sample Size Calculations: Practical Methods for Engineers and Scientists*. Fairport Harbor, OH: Mathews Malnar and Bailey.

Also see

- [PSS-2] **power** — Power and sample-size analysis for hypothesis tests
- [PSS-2] **power, graph** — Graph results from the power command
- [PSS-2] **power, table** — Produce table of results from the power command
- [PSS-5] **Glossary**
- [R] **sdtest** — Variance-comparison tests

Stata, Stata Press, and Mata are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2023 StataCorp LLC, College Station, TX, USA. All rights reserved.

